Autocad VBA初级教程 （强烈推荐）
Autocad VBA初级教程(第一课:入门)
1.为什么要写这个教程
市面上ACAD VBA的书不多，它的帮助是英文版的，很多人看不懂。其实我转行已经好几年了，而且手艺也慢慢生疏了，写个教程对自己来说也是一次复习。
2.什么是Autocad VBA？
VBA是Visual Basic for Applications的英文缩写，它是一个功能强大的开发工具，学好VBA可以成倍甚至成百、成万倍提高工作效率，在工作中，有很多任务仅用ACAD命令不可能完成的，只要学好VBA就可以做到，相信到时候您一定会得到同事的佩服、老板的器重。
3、VBA有多难？
相信大家都知道Basic是的含义。应该承认，我的水平还不高，错误之处在所难免，如果大家发现错误一定要提出批评，以便及时更正。
4、怎样学习VBA？
介绍大家一个学习公式：信心+恒心=开心。仔细阅读本教程，完成例题，在学习的过程中一定要多思考，多想一些是什么、为什么。本教程将陆续发布在CAD世界论坛上，您不需要付费就可以学习。本作者在此郑重承诺：关于本教程中有任何疑问，可以跟贴提问，只要有时间，本人一定会耐心解答。我不会发到任何人的邮箱中，您自己在论坛上找就可以了，请不要再向我索要这份教程。
5、现在我们开始编写第一个程序：画一百个同心圆
第一步：复制下面的代码
第二步：在模型空间按快捷键Alt+F8，出现宏窗口
第三步：在宏名称中填写C100，点“创建”、“确定”
第四步：在Sub c100()和End Sub之间粘贴代码
第五步：回到模型空间，再次按Alt+F8，点击“运行”
Sub c100()
Dim cc(0 To 2) As Double '声明坐标变量
cc(0) = 1000 '定义圆心座标
cc(1) = 1000
cc(2) = 0
For i = 1 To 1000 Step 10 '开始循环
  Call ThisDrawing.ModelSpace.AddCircle(cc, i * 10) '画圆
Next i
End Sub
也许您还看不懂上面的代码，这没有关系，只要能把同心画出来就可以了，祝您成功。
Autocad VBA初级教程 (第二课 编程基础)
本课主要任务是对上一课的例程进行详细分析
下面是源码：
Sub c100()
Dim cc(0 To 2) As Double '声明坐标变量
cc(0) = 1000 '定义圆心座标
cc(1) = 1000
cc(2) = 0
For i = 1 To 1000 Step 10 '开始循环
  Call ThisDrawing.ModelSpace.AddCircle(cc, i * 10) '画圆
Next i
End Sub
先看第一行和最后一行：
Sub C100()
……
End Sub
C100是宏的名称，也叫过程名称，当用户执行C100时程序将运行sub　和end sub之间的所有指令。
第二行：
Dim cc(0 To 2) As Double '声明坐标变量
后半段“'声明坐标变量”自动变为绿色字体，它是代码语句的注释，它不会影响程序运行，它的作用是告诉阅读者程序员的想法。对于简单的程序，一般不需要写注释，如果要编写非常复杂的程序，最好要多加注释，越详细越好，对于程序员来说，这是一个好习惯。
电脑真正编译执行的是这条语句：Dim cc(0 To 2) As Double
它的作用就是声明变量。
Dim是一条语句，可以理解为计算机指令。
它的语法：Dim变量名 As 数据类型
本例中变量名为CC，而括号中的0 to 2声明这个CC是一个数组，这个数组有三个元素：CC(0)、CC(1)、CC(2)，如果改为CC(1 to 3)，则三个元素是CC(1)、CC(2)、CC(3)，有了这个数组，就可以把坐标数值放到这个变量之中。
Double是数据类型中的一种。ACAD中一般需要定义坐标时就用这个数据类型。在ACAD中数据类型的有很多，下面两个是比较常用的数据类型，初学者要有所理解。
Long（长整型），其范围从 -2,147,483,648 到 2,147,483,647。
Variant  它是那些没被显式声明为其他类型变量的数据类型，可以理解为一种通用的数据类型，这是最常用的。
下面三条语句
cc(0) = 1000 '定义圆心座标
cc(1) = 1000
cc(2) = 0
它们的作用是给CC变量的每一个元素赋,值其顺序是X、Y、Z坐标。
For i = 1 To 1000 Step 10 '开始循环
……
Next i  '结束循环
这两条语句的作用是循环运行指令，每循环一次，i值要增加10，当i加到 1000时，结束循环。
i也是一个变量，虽然没有声明i变量，程序还是认可的，VB不是C语言，每用一个变量都要声明，不声明就会报错。简单是简单了，这样做也有坏处，如果不小心打错了一个字母，程序不会报错，如果程序很长，那就会出现一些意想不到的错误。
step后面的数值就是每次循环时增加的数值，step后也可以用负值。
例如：For i =1000 To 1 Step -10 
很多情况下，后面可以不加step 10
如：For i=1 to 100，它的作用是每循环一次i值就增加1
Next i语句必须出现在需要结束循环的位置，不然程序没法运行。
下面看画圆命令：
Call ThisDrawing.ModelSpace.AddCircle(cc, i * 10)
Call语句的作用是调用其他过程或者方法。
ThisDrawing.ModelSpace是指当前CAD文档的模型空间
AddCircle是画圆方法
Addcicle方法需要两个参数：圆心和半径
CC就是圆心坐标，i*10就是圆的半径，本例中，这些圆的半径分别是10、110、210、310……
本课到此结束，下面请完成一道思考题：
1.以(4,2)为圆心，画5个同心圆，其半径为1-5
Autocad VBA初级教程 (第三课 编程基础二)
有人提出了下面的问题：
绘制三维多段线时X、Y值在屏幕上用鼠标选取，Z值用键盘输入
本课将讲解这个问题。
为了简化程序，这里用多条直线来代替多段线。以下是源码：
Sub myl()
Dim p1 As Variant '申明端点坐标
Dim p2 As Variant
p1 = ThisDrawing.Utility.GetPoint(, "输入点:") '获取点坐标
z = ThisDrawing.Utility.GetReal("Z坐标:") '用户输入Z坐标值
p1(2) = z '将Z坐标值赋予点坐标中
On Error GoTo Err_Control '出错陷井
Do '开始循环
  p2 = ThisDrawing.Utility.GetPoint(p1, vbCr & "输入下一点:") '获取下一个点的坐标
  z = ThisDrawing.Utility.GetReal("Z坐标:") '用户输入Z坐标值
  p2(2) = z '将Z坐标值赋予点坐标中
  Call ThisDrawing.ModelSpace.AddLine(p1, p2) '画直线
  p1 = p2 '将第二点的端点保存为下一条直线的第一个端点坐标
Loop
Err_Control:
End Sub
先谈一下本程序的设计思路：
1、获取第一点坐标
2、输入第一点Z坐标
3、获取第二点坐标
4、输入第二点Z坐标
5、以第一、二点为端点，画直线
6、下一条线的第一点=这条线的第二点
7、回到第3步进行循环
如果用户没有输入坐标或Z值，则程序结束。
首先看以下两条语句：
p1 = ThisDrawing.Utility.GetPoint(, "输入点:") ‘获取点坐标
……
p2 = ThisDrawing.Utility.GetPoint(p1, vbCr & "输入下一点:") '获取下一个点的坐标
这两条语句的作用是由用户输入点用鼠标选取点坐标，并把坐标值赋给p1、p2两个变量。ThisDrawing.Utility.GetPoint（）在ACAD中这是最常用的方法之一，它需要两个参数，在逗号前面的参数应该是一个点坐标，它的作用是在屏幕上画一条线，前一个端点位于点坐标位置，后一个端点跟随鼠标移动，逗号之前可以什么都不填，这时没有线条会跟随鼠标移动，但逗号必须保留。
逗号后面使用一串字符，程序在命令行显示这串字符，这不难理解。
VbCr通常代表一个回车符，而在这个语句中，它的作用是在命令行不显示“命令：”
&的作用是连接字符。举例：
“爱我中华 ”&”抵制日货 ”&”从我做起”
z = ThisDrawing.Utility.GetReal("Z坐标:") '用户输入Z坐标值
由用户输入一个实数
On Error GoTo Err_Control '出错陷井
……
Err_Control:
On Error是出错陷井语句，在程序出错时将执行On Error 后面的语句
GoTo Err_contorl 是程序跳转语句，它的作用是在程序中寻找Err_control:，并执行这一行后面的语句，本例中Err_Control:后就是结束宏，所以只要出现错误，程序中止。
Do '开始循环
……
Loop ‘结束循环
这个循环就历害了，它会无休止地进行循环，好在本例中已经有了一个出错陷井，当用户输入回车时，由于程序没有得到点或坐标，程序出错，跳出循环，中止程序。如果要人为控制跳出循环，可以在代码中用Exit Do语句跳出循环。在For 变量 和Next 变量之间如果要跳出循环，那么只要在循环体内加一个Exit for 就可以跳出循环，关于这方面的例程以后会讲到。
Call ThisDrawing.ModelSpace.AddLine(p1, p2) '画直线
画直线方法也是很常用的，它的两个参数是点坐标变量
本课到此结束，请做思考题：
连续画圆，每次要求用户输入圆心、半径，当用户不再输入圆心或半径时程序才退出
Autocad VBA初级教程 (第四课 程序的调试和保存)
人非圣贤，孰能无过，初学者在编写复杂程序时往往会出现一些意想不到的错误，所以程序的调试显得尤为重要，随着学习的深入，以后我们需要经常进行程序调试。事实上，对于那些资深程序员来说，调试程序也是一项不可或缺的重要工作。
首先，在程序输入阶段，应该充分利用VBA编辑器的智能功能。当你在写代码时，输入一些字母后，编辑器可以自动列出合适的语句、对象、函数供你选择，可以用上下键选择，然后按TAB键（它位于“Q”键左边）确认。当输入一个回车符后程序会自动对这条语句进行分析，如果出现错误就会提示。
我们经常碰到的麻烦是程序的运行结果和预计的不一样，一般我会这样做：首先要想一想可能是哪一个变量有问题，然后去监视这个变量(或表达式)，在程序合适的位置设置断点，这样可以使程序停下来看一看这个变量有没有按照我的设想在变化。下面我举一个简单的例子，先看源代码：
sub test()
for i=2 to 4 step 0.6
next i
end sub
这是一个非常简单的循环，每一次循环i便会增加0.6，当循环3次后i值就变为4.4，但问题是每一次循环时i值变为多少？
第一步：在菜单中选“调试”—“添加监视”，在表达试中填“i”，点击确定，这时你会看到临视窗口中会多一行。
第二步：把光标移到代码窗口中的“next i”行，按一下“F9”，于是每当程序运行到这里时就会暂停了。
好，一切就绪，请按F5执行程序，在监视窗口中C值立刻变为2，再按F5继续，C值为2.6，再按几次F5，直到程序结束，这样我们就成功监视了C值的变化。
第三步：在next i行再按一次F9，清除断点。监视的表达式的右键菜单选择“删除监视”。
另外，还可以用“逐语句”、“逐过程”、“运行到光标处”等方法进行调试，这些都在调试菜单中，操作比较简单，请读者自行领悟。
到目前为止，我们所做的工程都是“嵌入式工程”，它只是嵌入在当前的Autocad图形文件中， 以后打开这个文件时代码才会加载，如果别的dwg文件也要使用，那就需要把代码导出为.bas文件，供其他dwg文件导入。在VBA编辑器的“文件”菜单中有这两个功能，一试便知。
ACAD VBA还有一种工程叫“通用式工程”，只要进入ACAD就可以运行，程序可以在不同用户、不同的图形文件中共享，但是由于VBA功能太强，有时候会出现一些意想不到的事情，所以在学习阶段请暂时不要这样做。
本课结束，请做思考题；监视下列代码中的i和j的值，注意，此题虽然要监视2个变量，但是在代窗口中只要设置1个断点就足够了。
sub test()
for i=2 to 4 step 0.6
  for j=-5 to 2 step 5.5  
  next j
next i
end sub

Autocad VBA初级教程 (第五课 画函数曲线)
先画一组下图抛物线。
[attach]48[/attach] 
下面是源码：
Sub myl()
Dim p(0 To 49) As Double '定义点坐标
Dim myl As Object '定义引用曲线对象变量
co = 15 '定义颜色
For a = 0.01 To 1 Step 0.02 '开始循环画抛物线
  For i = -24 To 24 Step 2 '开始画多段线
    j = i + 24  '确定数组元素
    p(j) = i '横坐标
    p(j + 1) = a * p(j) * p(j) / 10 '纵坐标
  Next i '至此p(0)-p(40)所有元素已定义，结束循环
  Set myl = ThisDrawing.ModelSpace.AddLightWeightPolyline(p) '画多段线
  myl.Color = co '设置颜色属性
  co = co + 1 '改变颜色，供下次定义曲线颜色
Next a
End sub
为了鼓励大家积极思考，从本课开始，我不再解释每一条语句的作用，只对以前没有提过的语句进行一些解释，也许你一时很难明白，建议用上一课提到的跟踪变量、添加断点的办法领悟每一条语句的作用，如果有问题不懂请跟贴提问。
在跟踪变量p时请在跟踪窗口中单击变量p前的+号，这样可以看清数组p中每一个元素的变化。
ACAD没有现成的画抛物线命令，我们只能用程序编写多段线画近似抛物线。理论上，抛物线的X值可以是无限小、无限大，这里取值范围在正负24之间。
程序第二行：Dim myl As Object '定义引用曲线对象变量
Object也是一种变量类型，它可以把变量定义为对象，本例中myl变量将引用多段线，所以要定义为Objet类型。
看画多段线命令：
Set myl = ThisDrawing.ModelSpace.AddLightWeightPolyline(p) '画多段线
其中括号中的p是一个数组，这个数组的元素数必须是偶数，每两个元数作为一个点坐标。
等号前面部分“Set myl”的作用就将myl变量去引用画好的多段线。
myl.Color = co '设置颜色属性。在ACAD中，颜色可以用数字表示，本例中co会增值，这样就会有五彩缤纷的效果。
本课第二张图：正弦曲线，下面是源码：
Sub sinl()
Dim p(0 To 719) As Double '定义点坐标
For i = 0 To 718 Step 2 '开始画多段线
    p(i) = i * 2 * 3.1415926535897 / 360 '横坐标
    p(i + 1) = 2 * Sin(p(i)) '纵坐标
Next i
ThisDrawing.ModelSpace.AddLightWeightPolyline (p) '画多段线
ZoomExtents '显示整个图形
End Sub
p(i) = i * 2 * 3.1415926535897 / 360 '横坐标
横坐标表示角度，后面表达式的作用是把角度转化弧度
ZoomExtents语句是缩放命令，它的作用是显示整个图形，消除图形以外的区域
本课思考题：画一条抛物线：y=0.5*x*x+3，其中X取值范围在正负50之间
Autocad VBA初级教程 (第六课 数据类型的转换)
上一节课我们用一个简单的公式把角度转化为弧度，这样做便于大家理解。不过VBA中有现成的方法可以转换数据类型。
我们举例说明：
jd = ThisDrawing.Utility.AngleToReal(30, 0)
这个表达式把角度30度转化为弧度，结果是.523598775598299。
AngleToReal需要两个参数，前面是表示要转换角度的数字，而后面一个参数可以取值为0-4之间的整数，有如下意义：
0：十进制角度；1：度分秒格式；2：梯度；3：弧度；4：测地单位
例：id= ThisDrawing.Utility.AngleToReal("62d30' 10""", 1)
这个表达式计算62度30分10秒的弧度
再看将字符串转换为实数的方法：DistanceToReal
需要两个参数，前一个参数是表示数值的字符串，后面可以取值1-5，表示数据格式，有如下意义：
1：科学计数；2：十进制；3：工程计数——英尺加英寸；4：建筑计数——英尺加分数英寸；5：分数格式。
例：以下表达式得到一个12.5的实数
temp1 = ThisDrawing.Utility.DistanceToReal("1.25E+01", 1)
temp2 = ThisDrawing.Utility.DistanceToReal("12.5", 2)
temp3 = ThisDrawing.Utility.DistanceToReal("12 1/2", 5)
而realtostring方法正好相反，它把一个实数转换为字符串。它需要3个参数
第一个参数是一个实数，第二个参数表示数据格式，含义同上，最后一个参数表示精确到几位小数。
temp1= ThisDrawing.Utility.RealToString(12.5, 1, 3)
得到这个字符串：“1.250E+01”，
下面介绍一些数型转换函数：
Cint，获得一个整数，例：Cint(3.14159) ，得到3
Cvar，获得一个Variant类型的数值，例：Cvar("123" & "00")，得到”12300”
Cdate，转换为date数据类型，例：MyShortTime = CDate("11:13:14 AM")
下面的代码可以写出一串数字，从000-099。
Sub test()
Dim add0 As String
Dim text As String
Dim p(0 To 2) As Double
p(1) = 0 'Y坐标为0
p(2) = 0 'Z坐标为0
For i = 0 To 99 '开始循环
  If i < 10 Then '如果小于10
    add0 = "00" '需要加00
  Else '否则
    add0 = "0" '需要加0
  End If
  text = add0 & CStr(i) '加零，并转换数据
  p(0) = i * 100 'X坐标
  Call ThisDrawing.ModelSpace.AddText(text, p, 4) '写字
  Next i
  
End Sub
重点解释条件判断语句：
If 条件表达式 Then 
……
Else
……
End if
如果满足条件那么程序往下执行，到else时不再往下执行，直接跳到End if后面
如果不满足条件，程序跳到else后往下运行。
  Call ThisDrawing.ModelSpace.AddText(text, p, 4) '写字
这是写单行文本，需要三个参数，分别是:写的内容、位置、字高
Autocad VBA初级教程 (第七课 写文字)
客观地说，ACAD写字功能不够历害，而用VBA可以使写字效率更高。比较正规的做法是把定义文字样式，用样式来控制文字的特性。我们还是用实例来学习，先看下面一段代码，它的作用是先创建一个文字样式，然后用这个文字样式写一段多行文本。
Sub txt()
Dim mytxt As AcadTextStyle '定义mytxt变量为文本样式
Dim p(0 To 2) As Double '定义坐标变量
p(0) = 100: p(1) = 100: p(2) = 0 '坐标赋值
Set mytxt = ThisDrawing.TextStyles.Add("mytxt") '添加mytxt样式
mytxt.fontFile = "c:\windows\fonts\simfang.ttf" '设置字体文件为仿宋体
mytxt.Height = 100 '字高
mytxt.Width = 0.8 '宽高比
mytxt.ObliqueAngle = ThisDrawing.Utility.AngleToReal(3, 0) '倾斜角度（需转为弧度）
ThisDrawing.ActiveTextStyle = mytxt '将当前文字样式设置为mytxt
Set txtobj = ThisDrawing.ModelSpace.AddMText(p, 1400, "{做到老，学到老}\P" & "此心自光明正大，过人远矣")
txtobj.LineSpacingFactor = 2 '指定行间距
txtobj.AttachmentPoint = 3 '右对齐（1为左对齐，2为居中）
End Sub
我们看这条语句
Set mytxt = ThisDrawing.TextStyles.Add("mytxt") 
添加文本样式并赋值给mytxt变量，只需要一个参数：文本样式名
fontfile、height、width、ObliqueAngle是文本样式最常用的属性
Call ThisDrawing.ModelSpace.AddMText(p, 1400, "{做到老，学到老}\P" & "此心自光明正大，过人远矣")
这条语句是写文本，需要三个参数。第一个参数p是坐标，1400是宽度，最后一个参数是文本内容，其中\p是一个回车符
扩大字符间距用\T数字，例：\T3abc，使文字abc的间距扩大3 部，n取值范围是0.75-3
在论坛中有一个经常被同好提及的问题：如何使用文字叠加。举例说明：123\S+0.12^-0.34
\S是格式字符，^是分隔符，前面的数字在上，后面的数字在下。
\C是颜色格式字符，C后面跟一个数字表示颜色
\A是对齐方式，\A0，\A1,\A2分别表示底部对齐、中间对齐和顶部对齐
Autocad VBA初级教程 (第八课:图层操作)
先简单介绍两条命令：
1、这条语句可以建立图层：
ThisDrawing.Layers.Add("新建图层")
在括号中填写图层的名称。
2、设置为当前的图层
ThisDrawing.ActiveLayer=图层对象
注意，等号右边的变量不能用图层名称，必须使用一个有效的图层变量
以下一些属性在图层比较常用：
LayerOn 打开关闭
Freeze 冻结
Lock锁定
Color 颜色
Linetype 线型
看一个例题：
1、先在已有的图层中寻找一个名为“新建图层”的图层
2、如果找到这个图层，显示该图层的信息，并提示用户是否需要设置为当前图层，如果用户确认，则设置为当前图层。
3、如果图层没有找到，新建一个名为“新建图层”的图层，设置为黄色，HIDDEN线型，并把这个图层设置为当前图层
Sub mylay()
Dim lay0 As AcadLayer '定义作为图层的变量
Dim lay1 As AcadLayer
findlay = 0 '寻找图层的结果的变量，0没有找到，1找到
For Each lay0 In ThisDrawing.Layers '在所有的图层中进行循环
  If lay0.Name = "新建图层" Then '如果找到图层名
    findlay = 1 '把变量改为1标志着图层已经找到
    msgstr = lay0.Name + "已经存在" + vbCrLf
    msgstr = msgstr + "图层状态：" + IIf(lay0.LayerOn = True, "打开", "关闭") + vbCrLf
    msgstr = msgstr + "图层" + IIf(lay0.Freeze = True, "已经", "没有") + "冻结" + vbCrLf
    msgstr = msgstr + "图层" + IIf(lay0.Lock = True, "已经", "没有") + "锁定" + vbCrLf
    msgstr = msgstr + "图层颜色号：" + CStr(lay0.Color) + vbCrLf
    msgstr = msgstr + "图层线型：" + lay0.Linetype + vbCrLf
    msgstr = msgstr + "图层线宽：" + CStr(lay0.Lineweight) + vbCrLf
    msgstr = msgstr + "打印开关" + IIf(lay0.Plottable = False, "关闭", "打开") + vbCrLf + vbCrLf
    msgstr = msgstr + "是否设置为当前图层？"
    If MsgBox(msgstr, 1) = 1 Then '如果用户点击确定
       If Not lay0.LayerOn Then lay0.LayerOn = True '打开
       ThisDrawing.ActiveLayer = lay0 '把当前图层设为已经存在的图层
    End If
    Exit For '结束寻找
  End If
Next lay0
If findlay = 0 Then '没有找到图层
  Set lay1 = ThisDrawing.Layers.Add("新建图层") '增加一个名为“临时图层”的图层
  lay1.Color = 2 '图层设置为黄色 
  ltfind = 0 '找到线型的标志，0没有找到，1找到
  For Each entry In ThisDrawing.Linetypes '在现有的线型中进行循环
    If StrComp(entry.Name, "HIDDEN") = 0 Then '如果线型名为"HIDDEN"
      ltfind = 1 '标志为已找到线型
      Exit For '退出循环
    End If
  Next entry '结束循环
  If ltfind = 0 Then '没有找到线型
    ThisDrawing.Linetypes.Load "HIDDEN", "acadiso.lin" '加载线型
  End If
  lay1.Linetype = "HIDDEN" '设置线型
  ThisDrawing.ActiveLayer = lay1 '将当前图层设置为新建图层
End If
End Sub
在寻找图时时我们用到for each……next 语句
它的语法是这样的：
For Each 变量 In 数组或集合对象
……
exit for 
……
next 变量
它的作用是在数组或集合对象中进行循环，每循环一次，变量就成为数组或集合对象中的一个元素。本例在所有的图层对象中进行循环，每循环一次layo变量就代表一个图层
在循环体中遇到exit for 语句则退出循环，如果没有 exit for，循环将在所有的元素都操作一遍后结束。
If lay0.Name = "新建图层" Then
lay0.name代表这处图层的图层名
IIf(lay0.LayerOn = True, "打开", "关闭")
这是一个简单判断语句，语法如下：
iif（判断表达式，返回值1，返回值2）
当判断表达式成立，函数值=返回值1，如果表达式不成立，函数值=2
MsgBox(msgstr, 1) 
Mgbox显示一个对话框，第一个参数是对话框显示的内容
第二个参数可以控制对话框上的按钮。
0 只有确认按钮
1 确认、取消
2 终止、重试、忽略
3 是、否、取消
4 是、否
MsgBox获得值如下：
确认：1
取消：2
终止：3
重试：4
忽略：5
是：6
否7
初学者不需要死记硬背，能有所了解就行了
ACAD图层中最麻烦的就是线型问题了，本例先寻找一个HIDDEN线型，如果找不到就加载这个线型，用这条语句：
ThisDrawing.Linetypes.Load "HIDDEN", "acadiso.lin" 
ThisDrawing.Linetypes.Load后需要两个参数，一个是线型的名称，另外一个是线型文件的名称。
Autocad VBA初级教程 (第九课:创建选择集)
1.在创建对象的同时可以直接引用.以前的例程中已经做过多次了,现在复习一下,看例程:先随机画300个圆,在画圆时直接引用,然后再把这些圆根本大小修改颜色.
Sub c300()

Dim myselect(0 To 300) As AcadEntity '定义选择集数组
Dim pp(0 To 2) As Double '圆心坐标
For i = 0 To 300 '循环300次
pp(0) = 3000 * Rnd: pp(1) = 3000 * Rnd: pp(2) = 0 '设置圆心坐标
Set myselect(i) = ThisDrawing.ModelSpace.AddCircle(pp, Rnd * 30 + 1) '画不同大小的圆
Next i

For i = 1 To 300
If myselect(i).Radius > 10 Then '判断圆的直径是否大于10
 myselect(i).color = Int(255 * Rnd + 1) '大圆颜色改为随机数
Else
 myselect(i).color = 0 '小圆改为白色
End If
Next i

ZoomExtents '缩放到显示全部对象
End Sub
pp(0) = 3000 * Rnd: pp(1) = 3000 * Rnd: pp(2) = 0
这一行实际上应该是三条语句,用三行合并为一行,用冒号分开
rnd是随机数函数,它的数值为0-1之间的小数,3000*rnd得到的数值就是在0-3000之间的随机数
Set myselect(i) = ThisDrawing.ModelSpace.AddCircle(pp, Rnd * 30 + 1)
这条语句的作用是以pp点坐标为圆心,画一个圆,半径是1-30之间的随机数,赋值给myselect选择集.
2.提标用户在屏幕中选取
选择语句这样写:Set sset = ThisDrawing.SelectionSets.Add("ss1"),其中”ss1”是一个选择集名称,这个参数可以随意写一个,注意不要重复就可以了.
下面的例程是让用户选择对象,然后把选中的对象改为绿色,最后把选择集删除
Sub mysel()

Dim sset As AcadSelectionSet '定义选择集对象
Dim element As AcadEntity '定义选择集中的元素对象
Set sset = ThisDrawing.SelectionSets.Add("ss1") '新建一个选择集
sset.SelectOnScreen '提示用户选择
For Each element In sset '在选择集中进行循环
  element.color = acGreen '改为绿色
Next
sset.Delete '删除选择集
End Sub

3.选择全部对象
用select方法,参数为acSelectionSetAll ,看例程,这个程序选择全部对象,显示选中的对象,并计算对象数.
Sub allsel()
Dim sel1 As AcadSelectionSet '定义选择集对象
Set sel1 = ThisDrawing.SelectionSets.Add("s") '新建一个选择集
Call sel1.Select(acSelectionSetAll) '全部选中
sel1.Highlight (True) '显示选择的对象
sco= sel1.Count '计算选择集中的对象数
MsgBox "选中对象数:" & CStr(sco) '显示对话框
End Sub
3.运用select方法
上面的例题已经运用了select方法,下面讲一下select的5种选择方式:
1:择全部对象(acselectionsetall)
2.选择上次创建的对象(acselectionsetlast)
3.选择上次选择的对象(acselectionsetprevious)
4.选择矩形窗口内对象(acselectionsetwindow)
5.选择矩形窗口内以及与边界相交的对象(acselectionsetcrossing)
还是看代码来学习.其中选择语句是:
Call sel1.Select(Mode, p1, p2)
Mode已经定义为5,也就是选择矩形窗口内以及与边界相交的对象,p1和p2是两个点坐标,
Sub selnew()
Dim sel1 As AcadSelectionSet '定义选择集对象
Dim p1(0 To 2) As Double '坐标1
Dim p2(0 To 2) As Double '坐标2
p1(0) = 0: p1(1) = 0: p1(2) = 0 '设置坐标1
p2(0) = 300: p2(1) = 300: p2(2) = 0 '设置坐标1
Mode = 5 '把选择模式存入mode变量中
Set sel1 = ThisDrawing.SelectionSets.Add("sel3") '新建一个选择集
Call sel1.Select(Mode, p1, p2) '选择对象
sel1.Highlight (ture) '显示已选中的对象
End Sub

Autocad VBA初级教程 (第十课:画多段线和样条线)

画二维多段线语句这样写：
set lobj= ThisDrawing.ModelSpace.AddLightweightPolyline(fitpoint)
AddLightweightPolyline后面需一个参数，存放顶点坐标的数组
画三维多段线语句这样写：
Set lobj = ThisDrawing.ModelSpace.Add3DPoly(fitpoint)
Add3dpoly后面需一个参数，就是顶点坐标数组
画二维样条线语句这样写：
Set lobj = ThisDrawing.ModelSpace.AddSpline(fitPoints,startT, endT)
Addspline后需要三个参数，第一个是拟合点数组，后面两个分别是起点切向和终点切向。
下面看例题。这个程序是第三课例程的改进版。原题是这样的：
绘制三维多段线时X、Y值在屏幕上用鼠标选取，Z值用键盘输入。
细心的读者是否还记得这个小程序已经在第三课的回贴中发布了。先分析一下编程细路：
用动态数组存放多段线的所有顶点坐标，获得新坐标后就把前一次画的多段线删除，画出多段线，并将线段第二点的端点保存为下一条线段的第一个端点坐标，以下是源码：
Sub myl()

Dim p1 As Variant '申明端点坐标
Dim p2 As Variant
Dim l() As Double '声明一个动态数组
Dim templ As Object

p1 = ThisDrawing.Utility.GetPoint(, "输入点:") '获取点坐标
z = ThisDrawing.Utility.GetReal("Z坐标:") '用户输入Z坐标值
p1(2) = z '将Z坐标值赋予点坐标中
ReDim l(0 To 2) '定义动态数组
l(0) = p1(0)
l(1) = p1(1)
l(2) = z

On Error GoTo Err_Control '出错陷井
Do '开始循环
  p2 = ThisDrawing.Utility.GetPoint(p1, vbCr & "输入下一点:") '获取下一个点的坐标
  z = ThisDrawing.Utility.GetReal("Z坐标:") '用户输入Z坐标值
  p2(2) = z '将Z坐标值赋予点坐标中 
  lub = UBound(l) '获取当前l数组中元的元素个数
  ReDim Preserve l(lub + 3)
  For i = 1 To 3
    l(lub + i) = p2(i - 1)
  Next i
  If lub > 3 Then
    templ.Delete '删除前一次画的多段线
  End If
  Set templ = ThisDrawing.ModelSpace.Add3DPoly(l) '画多段线
  p1 = p2 '将第二点的端点保存为下一条直线的第一个端点坐标
Loop
Err_Control:

End Sub
我们学一学动态数组，这是非常有用的。所谓动态数组，简单讲就是数组元素可以改变的特殊数组。
这样定义数组：Dim l( ) As Double 
赋值语句：
ReDim l(0 To 2) 
l(0) = p1(0)
l(1) = p1(1)
l(2) = z
重新定义数组元素语句：
  lub = UBound(l) '先要获取当前l数组中元的元素个数,用ubount函数计算。
  ReDim Preserve l(lub + 3)
重定义时数组中要加三个元素，用以存放坐标值，所以定义元素为lub+3。给数值赋值的语句是一样的。
再看画多段线语句：
Set templ = ThisDrawing.ModelSpace.Add3DPoly(l) '画多段线
在最后面括号中的l就是存放点坐标的动态数组，这条语句在画多段线的同时把多段线作为对象赋值给templ 变量，这样下面的删除语句就可以直接引用这个变量了。
删除语句：
templ.Delete
因为已经知道对象是templ，所以只要在对象后面用delete方法就可以了。
下面的例程更加实用些，它的功能是把样条线转化为多段线。编程思路：用户选择一根样条线，计算样条线的拟合点，然后把所有的拟合点存放到数组中，最后用这个数组画多段线。
Sub sp2pl()

Dim getsp As Object ‘获取样条线的变量
Dim newl() As Double ‘多段线数组
Dim p1 As Variant ‘获得拟合点点坐标
ThisDrawing.Utility.GetEntity getsp, po, "本程序将样条曲线转为多段线。请选择样条曲线"
sumctrl = getsp.NumberOfControlPoints ‘计算样条线中一共有多少拟合点
ReDim newl(0 To sumctrl * 3 - 1) ‘重定义数组
  For i = 0 To sumctrl - 1 ‘开始循环，
  p1 = getsp.GetControlPoint(i) ‘把拟合点坐标存到p1变量中
      For j = 0 To 2
    newl(i * 3 + j) = p1(j)
  Next j

Next i
Set templ = ThisDrawing.ModelSpace.Add3DPoly(newl) ‘画样条线
End Sub

下面的语句是让用户选择样条线：
ThisDrawing.Utility.GetEntity getsp, po, "本程序将样条曲线转为多段线。请选择样条曲线"
ThisDrawing.Utility.GetEntity 后面需要三个参数：
第一个getsp用于存放获得用户选择的对象变量，第二个po是用户鼠标点下时的位置，第三个是一段字符串，显示在提示栏。
Autocad VBA初级教程 (第十一课:动画基础)

    说实话，用ACAD玩动画绝对比不上专业的三维动画软件，不过这年头自称精通CAD的工程师太多了，而学会用ACAD玩动画一定能提高您的竞争实力， 想象一下您用ACAD动画向老板演示零件的装配方法……
    下面的例题是一个最简单的动画，由用户选择一个要移动的对象，然后定位起点和终点，程序会慢慢地把对象移动。
    移动方法：object.move 起点坐标,端点坐标
Sub testmove()

Dim p0 As Variant       '起点坐标
Dim p1 As Variant       '终点坐标
Dim pc As Variant       '移动时起点坐标
Dim pe As Variant       '移动时终点坐标
Dim movx As Variant     'x轴增量
Dim movy As Variant     'y轴增量
Dim getobj As Object    '移动对象
Dim movtimes As Integer '移动次数
ThisDrawing.Utility.GetEntity getobj, po, "请选择移动对象"
p0 = ThisDrawing.Utility.GetPoint(, "起点:")
p1 = ThisDrawing.Utility.GetPoint(p0, "终点:")
pe = p0
pc = p0

motimes = 3000
movx = (p1(0) - p0(0)) / motimes
movy = (p1(1) - p0(1)) / motimes
For i = 1 To motimes
  pe(0) = pc(0) + movx
  pe(1) = pc(1) + movy
  getobj.Move pc, pe    '移动一段
  getobj.Update         '更新对象
Next

End Sub

先让用户选择移动的对象、起点、终点，motimes动画是设置动画帧数，这个例程是让对象移动3000小段，movx和movy是每一段移动的距离，然后进行一个3000次的循环，每一次循环移动一小段距离。
看第二个例题：做一个小轮子在一条山坡上运动的动画，山坡为正弦曲线。请注意，应该用偏移法获得轮圆心的轨迹才是正确的。编程思路：先画好圆和横轴，然后画山坡，偏移获得移动轨迹线，放到隐藏图层中，然后根据轨迹线移动圆和轴，在移动之前先旋转轴线。
旋转方法：object. rotate 基点,角度
偏移方法： object.offset(偏移量)
Sub moveball()

Dim ccball As Variant '圆
Dim ccline As Variant '圆轴
Dim cclinep1(0 To 2) As Double '圆轴端点1
Dim cclinep2(0 To 2) As Double '圆轴端点2
Dim cc(0 To 2) As Double '圆心
Dim hill As Variant '山坡线
Dim moveline As Variant '移动轨迹线
Dim lay1 As AcadLayer '放轨迹线的隐藏图层
Dim vpoints As Variant '轨迹点
Dim movep(0 To 2) As Double '移动目标点坐标
cclinep1(0) = -0.1: cclinep2(0) = 0.1 '定义圆轴坐标
Set ccline = ThisDrawing.ModelSpace.AddLine(cclinep1, cclinep2) '画直线
Set ccball = ThisDrawing.ModelSpace.AddCircle(cc, 0.1) '画半径为0.1的圆
Dim p(0 To 719) As Double   '申明正弦线顶点坐标
For i = 0 To 718 Step 2 '开始画多段线
    p(i) = i * 3.1415926535897 / 360  '横坐标
    p(i + 1) = Sin(p(i)) '纵坐标
Next i
Set hill = ThisDrawing.ModelSpace.AddLightWeightPolyline(p) '画正弦线即山坡曲线
hill.Update '显示山坡线
moveline = hill.Offset(-0.1) '球心运动轨迹线
vpoints = moveline(0).Coordinates '获得规迹点
Set lay1 = ThisDrawing.Layers.Add("hidelay") '创建名为"hidelay"的图层
lay1.LayerOn = False '关闭图层
moveline(0).Layer = "hidelay" '将轨迹线放到关闭的图层中
ZoomExtents '显示整个图形
For i = 0 To UBound(vpoints) - 1 Step 2
  movep(0) = vpoints(i) '计算移动的轨迹
  movep(1) = vpoints(i + 1)
  ccline.Rotate cc, 0.05 '旋转直线
  ccline.Move cc, movep '移动直线
  ccball.Move cc, movep '移动圆
  cc(0) = movep(0) '把当前位置作为下次移动的起点
  cc(1) = movep(1)
  For j = 1 To 50000 '这次循环的目的是让小球移动得慢一点,循环量应根据自已的电脑速度设置
   j = j * 1
  Next j
  ccline.Update '更新
Next i

End Sub


本课结束，请做思考题：在一条直线上画两个同样大小的圆，然后让这两个圆沿直线相向而行，到互换位置后停下来，尺寸自定
Autocad VBA初级教程 (第十二课:参数化设计基础)

简单地讲，参数化设计就是根据参数进行精确绘图，绘图所需要的参数也可以由用户手工输入。真正的参数化设计往往需要数据库操作，为了简化程序，把数据库部分放在以后的课程中详细讲解。
    本课的例程是画一个标准足球场。足球场长度90～120米，宽度45～90米，而红色标注的尺寸是程序默认的，绿色标注固定不变。



Sub court()
Dim courtlay As AcadLayer '定义球场图层
Dim ent As AcadEntity '镜像对象
Dim linep1(0 To 2) As Double '线条端点1
Dim linep2(0 To 2) As Double '线条端点2
Dim linep3(0 To 2) As Double '罚球弧端点1
Dim linep4(0 To 2) As Double '罚球弧端点2
Dim centerp As Variant '中心坐标
xjq = 11000 '小禁区尺寸
djq = 33000 '大禁区尺寸
fqd = 11000 '罚球点位置
fqr = 9150 '罚球弧半径
fqh = 14634.98 '罚球弧弦长
jqqr = 1000 '角球区半径
zqr = 9150 '中圈半径

On Error Resume Next
chang = ThisDrawing.Utility.GetReal("长度(90000～120000)<105000>")
If Err.Number <> 0 Then '用户输入的不是有效数字
  chang = 105000
  Err.Clear '清除错误
End If
kuan = ThisDrawing.Utility.GetReal("宽度(45000～90000)<68000>")
If Err.Number <> 0 Then
  kuan = 68000
End If

centerp = ThisDrawing.Utility.GetPoint(, "定位球场中心:")

Set courtlay = ThisDrawing.Layers.Add("足球场") '设置图层
ThisDrawing.ActiveLayer = courtlay '把当前图层设为足球场图层

'画小禁区
linep1(0) = centerp(0) + chang / 2
linep1(1) = centerp(1) + xjq / 2
linep2(0) = centerp(0) + chang / 2 - xjq / 2
linep2(1) = centerp(1) - xjq / 2
Call drawbox(linep1, linep2) '调用画矩形子程序

 

'画大禁区
linep1(0) = centerp(0) + chang / 2
linep1(1) = centerp(1) + djq / 2
linep2(0) = centerp(0) + chang / 2 - djq / 2
linep2(1) = centerp(1) - djq / 2
Call drawbox(linep1, linep2)


' 画罚球点
linep1(0) = centerp(0) + chang / 2 - fqd
linep1(1) = centerp(1)
Call ThisDrawing.ModelSpace.AddPoint(linep1)
'ThisDrawing.SetVariable "PDMODE", 32 '点样式
ThisDrawing.SetVariable "PDSIZE", 30 '点的尺寸

'画罚球弧，罚球弧圆心就是罚球点linep1
linep3(0) = centerp(0) + chang / 2 - djq / 2
linep3(1) = centerp(1) + fqh / 2
linep4(0) = linep3(0) '两个端点的x轴相同
linep4(1) = centerp(1) - fqh / 2
ang1 = ThisDrawing.Utility.AngleFromXAxis(linep1, linep3) '计算角度
ang2 = ThisDrawing.Utility.AngleFromXAxis(linep1, linep4)
Call ThisDrawing.ModelSpace.AddArc(linep1, zqr, ang1, ang2) '画弧


'角球弧
ang1 = ThisDrawing.Utility.AngleToReal(90, 0) '角度转换为弧度
ang2 = ThisDrawing.Utility.AngleToReal(180, 0)
linep1(0) = centerp(0) + chang / 2 '角球弧圆心
linep1(1) = centerp(1) - kuan / 2
Call ThisDrawing.ModelSpace.AddArc(linep1, jqqr, ang1, ang2) '画弧

ang1 = ThisDrawing.Utility.AngleToReal(270, 0)
linep1(1) = centerp(1) + kuan / 2
Call ThisDrawing.ModelSpace.AddArc(linep1, jqqr, ang2, ang1)

 

'镜像轴
linep1(0) = centerp(0)
linep1(1) = centerp(1) - kuan / 2
linep2(0) = centerp(0)
linep2(1) = centerp(1) + kuan / 2

'镜像
For Each ent In ThisDrawing.ModelSpace '所有模型空间的对象进行一次循环
  If ent.Layer = "足球场" Then '对象在"足球场"图层中
    ent.Mirror linep1, linep2 '镜像
  End If
Next ent

'画中线
Call ThisDrawing.ModelSpace.AddLine(linep1, linep2)

'画中圈
Call ThisDrawing.ModelSpace.AddCircle(centerp, zqr)

'画外框
linep1(0) = centerp(0) - chang / 2
linep1(1) = centerp(1) - kuan / 2
linep2(0) = centerp(0) + chang / 2
linep2(1) = centerp(1) + kuan / 2
Call drawbox(linep1, linep2)

ZoomExtents '显示整个图形

End Sub

Private Sub drawbox(p1, p2) '根据对角线坐标画矩形的子程序
Dim boxp(0 To 14) As Double

boxp(0) = p1(0)
boxp(1) = p1(1)

boxp(3) = p1(0)
boxp(4) = p2(1)

boxp(6) = p2(0)
boxp(7) = p2(1)

boxp(9) = p2(0)
boxp(10) = p1(1)

boxp(12) = p1(0)
boxp(13) = p1(1)

Call ThisDrawing.ModelSpace.AddPolyline(boxp)

End Sub
下面开始分析源码：

On Error Resume Next
chang = ThisDrawing.Utility.GetReal("长度(90～120)<10500>")
If Err.Number <> 0 Then '用户输入的不是有效数字
chang = 10500
Err.Clear '清除错误
End If

    这段代码的作用是要求用户输入一个足球场长度的数字，由于getreal只能输入数字，如果输入其他字符程序就会报错，所以先要用去掉错误提示：On Error Resume Next，虽然错误不再提示，但是出错代码会err.number改变，有兴趣的读者可以用变量跟踪的方法看看这个代码的数值。您只要记住，如果这个数字不是0，那么就是有错了，这时就可以把长度定为默认值，然后用Err.Clear语句把错误代码清零。


    在画小禁区的最后一行这样写：Call drawbox(linep1, linep2)

    Drawbox并不是vba提供的方法，它是一个带参数的子程序。由于画足球场要画好几次矩形，
而vba没有提供一个现成的画矩形方法，如果每次都用一长串代码画矩形是很麻烦的，所以需要把这些麻烦的代码写到一个子程序中，在需要时只有写一条调用语句就行了。这个子程序最后几行，从“Private Sub drawbox(p1, p2) ”开始，到end sub结束，p1,p2是参数，调用时也必须写两个参数：linep1、linep2。


ang1 = ThisDrawing.Utility.AngleFromXAxis(linep1, linep3) '计算角度
ang2 = ThisDrawing.Utility.AngleFromXAxis(linep1, linep4)
Call ThisDrawing.ModelSpace.AddArc(linep1, zqr, ang1, ang2) '画弧

    画圆用addarc方法，需要4个参数：圆心、半径、起始角度、结束角度。AngleFromXAxis用于计算角度，其参数需要两个点坐标

下面看镜像操作：
For Each ent In ThisDrawing.ModelSpace '所有模型空间的对象进行一次循环
  If ent.Layer = "足球场" Then '对象在"足球场"图层中
    ent.Mirror linep1, linep2 '镜像
  End If
Next ent

    本例只对“足球场”图层中的对象进行镜像，所以要对全部对象进行循环，判断对象的图层属性，只有位于“足球场”图层中的对象才作镜像。


本课思考题：
1、对本课的例程进行修改，当用户输入长、宽不在规定的范围时要求用户重新输入

2、设计一张简单的平面图，用户输入2个参数，其他尺寸写进程序中
如何应用VBA设AutoCAD定参数
AutoCAD一共有9个属于选项的对象，每一对象描述了选项对话框中的一个选项卡。这些对象提供了访问选项对话框中的所有注册信息项。你可以通过使用在这些对象中的属性来自定义AutoCAD设置。这些对象为 
PreferencesDisplay (显示参数) 
PreferencesDrafting (草图参数) 
PreferencesFiles (文件参数) 
PreferencesOpenSave (打开和保存参数) 
PreferencesOutput (输出参数) 
PreferencesProfile (配置参数) 
PreferencesSelection (选择参数) 
PreferencesSystem (系统参数) 
PreferencesUser (用户参数) 
这些对象可通过Preferences(参数)对象来访问。至于对Preferences对象的访问可使用Application(应用程序)对象中的Preferences属性。示例如下： 
访问Preferences(参数)对象 
Dim acadPref as AcadPreferences 
Set acadPref = ThisDrawing.Application.Preferences 
然后你可以使用Display、Drafting、Files、OpenSave、Output、Profile、Selection、System和User属性。例如，你可通过CursorSize属性控制十字光标的大小。 
设定十字光标为全屏 
本例设定十字光标为全屏。 
acadPref.Display.CursorSize = 100 
你可以通过应用程序使AutoCAD界面的某些外观启用或禁用。 
显示屏幕菜单和滚动条 
本例用DisplayScreenMenu和DisplayScrollBars属性使屏幕菜单可见，使滚动条不可见。 
acadPref.Display.DisplayScreenMenu = True 
acadPref.Display.DisplayScrollBars = False

    由于VBA集成在AutoCAD系统内部，因此用户必须先启动AutoCAD，然后才能进入VBA IDE环境。启动VBA的方式为：
    菜单：【Tools（工具）】→【Macor（宏）】→【Visual Basic Editor（VB编辑器）】 
    命令行：vbaide
    启动VB编辑器后，其常用界面如图37-4所示。
[image: image1.jpg]s

LT Tar BATAR

emawa

E37.4 VEAIDEMERRE




    各组成部分的说明如下：
    (1) 标题栏：显示应用程序名称和窗口控制按钮。
    (2) 菜单栏：通过选取各菜单项来完成相应的功能。
    (3) 工具栏：提供了对常用功能的快速调用。VBA IDE共提供了四个工具栏：“Standard（标准）”、“Edit（编辑）”、“UserForm（用户窗体）”和“Debug（调试）”，如图37-5所示。
    (4) 工程窗口：以树状结构显示主应用程序与窗体、模块及类模块的完整结构，用户可在工程窗口中直接选择窗体或模块，对其进行编辑。
    (5) 窗体窗口：构建应用程序界面，用户可以在窗体上添加各种控件。
    (6) 控件工具箱：用于设计视窗界面的各类控件。
    (7) 代码窗口：用于编写模块的过程代码。
    (8) 属性窗口：用于显示窗体上的某个控件的可见属性及其默认值，通过该属性窗口，还可以对可见属性的值进行直接编辑。

37.2.3 退出VBA IDE
    用户可选择菜单【File（文件）】→【Close and Return to AutoCAD（关闭并返回AutoCAD）】或单击窗口右上角的 按钮来退出VBA IDE并返回AutoCAD系统窗口。
实例1 最简单的VBA程序—“Hello.dvb”
    Step 1 创建新文件
    运行AutoCAD 2002系统，以“acadiso.dwt”为样板创建图形文件，并调用“vbaide”命令进入VBA环境；
    Step 2 创建窗体
    (1) 选择菜单【Insert（插入）】→【UserForm（用户窗体）】，编辑器将创建一个新的窗体，并显示在窗体窗口中。选择该窗体，然后在属性窗口中将“Caption”项改为“Draw Text”。
    (2) 在控件工具箱中单击 按钮，并在窗体的适当位置拖动鼠标，创建一个编辑框控件。
    (3) 在控件工具箱中单击 按钮，并在窗体的适当位置拖动鼠标，创建一个按钮控件。选择该控件后，在属性窗口中将“Caption”项改为“Click”。
    创建结果参见图37-6。

    Step 3 编写代码
    (1) 在窗体窗口中双击按钮控件，编辑器显示代码窗口，并提示用户输入代码，如图37-7所示。代码清单如下：
    Private Sub CommandButton1_Click()
    Dim TextObj As AcadText '定义文字对象变量
    Dim TextString As String '定义字符串变量
    Dim InsPnt(0 To 2) As Double '定义文字插入点数组变量
    Dim Height As Double '定义文字高度变量
    TextString = TextBox1.Text '字符串取值为编辑框中输入的文字
    '指定文字插入点位置和文字高度
    InsPnt(0) = 100: InsPnt(1) = 100: InsPnt(2) = 0
    Height = 15
   '在模型空间创建文字对象
    Set TextObj = ThisDrawing.ModelSpace.AddText(TextString, InsPnt, Height)
    TextObj.Color = acGreen '指定文字对象的颜色为绿色
    ZoomAll '缩放视图
    Unload Me '关闭窗体
    End Sub

    (2) 单击“Standard（标准）”工具栏中的 [image: image6.jpg]


按钮，以“Hello.dvb”为名保存该文件。
    Step 4 运行VBA程序
    (1) 单击“Standard（标准）”工具栏中的 [image: image7.jpg]


按钮运行该程序，系统将切换到AutoCAD窗口，并显示如图37-8所示的对话框。用户可在该对话框的编辑框中输入“Hello, VBA!”，并单击按钮，则将在当前图形中创建文字对象，结果如图37-9所示。

    实例说明
    如果用户退出VBA环境并返回AutoCAD系统窗口，则需要对该程序进行加载后才能运行。加载VBA程序的方式有如下几种：
    1. 选择菜单【Tools（工具）】→【Load Appcation…（加载应用程序）】，弹出“Load/Unload Applications（加载/卸载应用程序）”对话框。利用该对话框进行加载的过程与加载LISP程序相同。
    2. 选择菜单【Tools（工具）】→【Macro（宏）】→【Load Project…（加载工程）】，弹出“Open VBA Project（打开VBA工程）”对话框，用户可选择“Hello.dvb”文件并单击Open按钮进行加载。
    3. 选择菜单【Tools（工具）】→【Macro（宏）】→【VBA Manager…（VBA管理器）】，弹出“VBA Manager（VBA管理器）”对话框，如图37-10所示。

    该对话框中的“Drawing（图形）”下拉列表中显示了加载的所有图形文件。对于该列表中指定的图形文件，“Projects（工程）”列表显示了该文件中已加载的VBA程序，用户可单击 按钮载入其他的VBA绦颉?BR>

小 结
    本章主要介绍了AutoCAD ActiveX和VBA的概念和作用，并通过一个简单的实例讲述了在AutoCAD系统中开发VBA程序的过程。

如何应用VBA设定AutoCAD参数
　　AutoCAD一共有9个属于选项的对象，每一对象描述了选项对话框中的一个选项卡。这些对象提供了访问选项对话框中的所有注册信息项。你可以通过使用在这些对象中的属性来自定义AutoCAD设置。这些对象为 

PreferencesDisplay (显示参数) 

PreferencesDrafting (草图参数) 

PreferencesFiles (文件参数) 

PreferencesOpenSave (打开和保存参数) 

PreferencesOutput (输出参数) 

PreferencesProfile (配置参数) 

PreferencesSelection (选择参数) 

PreferencesSystem (系统参数) 

PreferencesUser (用户参数) 

这些对象可通过Preferences(参数)对象来访问。至于对Preferences对象的访问可使用Application(应用程序)对象中的Preferences属性。示例如下： 

访问Preferences(参数)对象 

Dim acadPref as AcadPreferences 

Set acadPref = ThisDrawing.Application.Preferences 

然后你可以使用Display、Drafting、Files、OpenSave、Output、Profile、Selection、System和User属性。例如，你可通过CursorSize属性控制十字光标的大小。 

设定十字光标为全屏 

本例设定十字光标为全屏。 

acadPref.Display.CursorSize = 100 

你可以通过应用程序使AutoCAD界面的某些外观启用或禁用。 

显示屏幕菜单和滚动条 

本例用DisplayScreenMenu和DisplayScrollBars属性使屏幕菜单可见，使滚动条不可见。 

acadPref.Display.DisplayScreenMenu = True 

acadPref.Display.DisplayScrollBars = False 
VBA编程如何使用Unicode字符、控制代码和特殊字符
　　使用Unicode字符、控制代码和特殊字符　 

可在文本字串中使用Unicode字符、控制代码和特殊字符以显示符号。(所有非文字字符必须用其ASCII等效值输入。) 

　 

Unicode 字符描述　： 

\U+00B0 角度符号 

\U+00B1 正负差符号 

\U+2205 直径标注符号 

除了使用Unicode字符显示特殊字符外，也可通过在文本字串中包括控制信息来指定特殊字符。使用一对百分比号(%%)以引入每一控制序列。 

该控制代码使用于标准AutoCAD字体和PostScript字体： 
%%nnn 
在 VB 或 VBA 字串中，以上例子要按以下输入 
Dim percent as Long 

percent = ASC("%") 

TextString = chr(percent) + chr(percent) + "nnn" 
这些控制代码只用于标准AutoCAD字体： 
控制代码描述：　 

%%o 切换上划线模式的开关 

%%u 切换下划线模式的开关 

%%d 绘制角度符号 

%%p 绘制正负差符号 

%%c 绘制直径标注符号 

%%% 绘制单个百分比号
避免VBA陷阱
　　有时，VBA使你避免危险 

作为Access-Office-VB ADVISOR 的技术编辑，我见过许多编程风格。我常常惊异于VBA程序员能够找到不同的方法来执行相同的任务，同时我也对VBA能够容忍如此差的技术而感到沮丧。VBA是一种相当宽容的语言，几乎能够运行一切那些糟糕的程序员所赋予它的任务。 

在我评价和测试本刊的文章之前，请首先关注五、六个我最经常遇到的编程错误。有一些编程手段对于每一个VBA代码都是必不可少的。 

毫无疑问，我所遇到的最常见的问题如下： 

[code]Dim rs1, rs2 As Recordset[/code] 

此处到底是什么类型的变量呢？在其他编程语言里，等价的语句将导致两个数据记录型变量，而在VBA中，你得到的是一个数据记录（rs2）和另一个变量（rs1）。虽然变量rs1运行时毫无错误，并且程序运行顺利，但是程序还是处于一种相当不确定的状态中。 

一个变量应当是参数可变的一种特定数据类型，它能够被赋予任何没有问题的数值，因此，变量rs1能够按照下面表述的形式接受任何数据记录型赋值： 

[code]Set rs1 = CurrentDb.OpenRecordset(“tblEmloyees”, dbOpenDynaset)[/code] 

问题在于每当代码中的rs1被执行时，VBA引擎都必须指明rs1变量的具体数值，在VBA决定如何处理该变量时，需要消耗额外的CPU时钟周期。而且，所有变量都会比等价的单个数据类型占据更多的内存。这些变量会降低应用程序的性能，使运行更加缓慢。 

此处还有一个例子，变量strMsg具有如下定义： 

[code]Dim strMsg[/code] 

任何包含strMsg的语句都将肯定比strMsg已经被定义为一个字符串的情况下要运行的缓慢。 

如果你对变量指定数据类型，对其进行定义的基础是要使你的应用程序运行的更快、更可靠。 

另外一个常见的问题是糟糕的代码缩排。下述代码段让人难以理解，如果将For…Next和If…End If循环进行仔细地缩排效果就会好得多：[code] 

For i=1 to 20 

If Condition1 Then 

Call SomeOperation1 

If Condition2 Then 

Call AnotherOperation 

End If 

End If 

Call AthirdOperation 

Next[/code] 

逻辑相关的语句，尤其是那些被当作独立单元进行执行的语句，应当缩进两格。如果你没有花一点时间指明一段逻辑相关的语句之间的逻辑关系，你可能会消耗更多的时间，因此进行适当的代码缩排是值得的。 

接下来的问题不太好用书面形式进行表述。因为刊物页面的幅宽有限，我不能将我所见到的长达470个字符的语句展示给你们。假设页面是标准的65个字符的幅宽，那么仅仅这个单句，就要换行8次，因此在编辑时，就要不得不向右滚动屏幕8次才能看清全句。尽管能够这样做，你也不可能一下把握全句的逻辑关系。显然，采用少量的分隔符将本长句划分为一些更短的句子，将便于管理。 

与长句类似，但并不象那么严重的一个问题如下： 

[code]Dim i As Integer , str As String , rs _ 

As Recordset , db As Database , db1 As Double[/code] 

在单行中进行许多定义，尤其当最后一个句子的末尾远远超出了编辑窗口的右端边界，将使得在大量的定义中查找某一个定义变得更加困难。 

此处还有一个常常易犯的编程问题： 

[code] 

If (Condition) Then 

rs.AddNew 

rs.FirstName = txtFirstName 

rs.LastName = txtLastName 

rs.Address = txtAddress 

rs.City = txtCity 

rs.State = txtState 

rs.ZipCode = txtZipCode 

rs..Update 

End If[/code] 

VBA提供特别的With…End with结构处理同一个对象的多个成分： 

[code] 

If (Condition) Then 

With rs 

.AddNew 

!FirstName = txtFirstName 

!LastName = txtLastName 

!Address = txtAddress 

!City = txtCity 

!State = txtState 

!ZipCode = txtZipCode 

.Update 

End With 

End If[/code] 

采用With…End With结构不仅可以使得表述简单，而且执行起来相当快。一旦VBA获得rs对象的句柄，那么With…End With代代码码块中的每一个成分将执行得更快。 

VBA的诸多优点中有一条是其对语句的使用方式具有较高的宽容性，在C或Pascal程序中可能会引起崩溃的代码也许能够在VBA中毫无错误地得以运行，但是这样的灵活性常常会付出代价。体系结构糟糕的VBA程序或其子程序比精心构造的程序运行得要缓慢，并且效率低下，另外调试和维护起来也更加困难。以上所有问题在矫正时都只需耗费少许的时间，但是花在学习和提高VBA代码编制技术上的时间将由更快的执行速度和更容易的日常维护来获得补偿。
如何用VBA创建匿名块？
　　在AutoCAD中，可以用VBA创建匿名块，那么如何创建呢？其实制作匿名块的方法和制作普通块的方法是一样的。如： 

Dim iPt(0 To 2) As Double 

iPt(0) = 0: iPt(1) = 0: iPt(2) = 0 

Dim BlockObj As AcadBlock 

Set BlockObj = ThisDrawing.Blocks.Add(iPt, "*U") 

上面的代码中在块集合生成了一个匿名块。看了上面的代码，是不是觉得制作匿名块其实很简单。但是如何获取刚创建的块的名称呢？其实匿名块的名称是以*+字母开头的，后面是从1开始递增的数字，那么如果是第一次创建匿名块，那名称就是“*U1”。下面的代码返回最后创建的匿名块： 

Public Function GetUNBlock() As AcadBlock 

Dim BlockObj As AcadBlock 

Dim n As Integer 

For Each BlockObj In ThisDrawing.Blocks 

If Left(BlockObj.Name, 1) = "*" Then 

If BlockObj.Name <> "*Model_Space" And Left(BlockObj.Name, 12) <> "*Paper_Space" Then 

If Mid(BlockObj.Name, 3) >= n Then 

n = Mid(BlockObj.Name, 3) 

Set GetUNBlock = BlockObj 

End If 

End If 

End If 

Next 

Set BlockObj = Nothing 

End Function
利用VBA开发AutoCAD
史大光　刘志海　高洪阁 

　　在绘制工程图形以及开发新产品、绘制零件图形时，为了区别零件的不同部位，经常用不同线型、不同宽度的线来表示，在某些场合中，零件的轮廓线经常要用粗实线来表示。一些工程技术人员为节省作图时间，经常将轮廓线先作成细实线，待确定无误之后，再将细实线加粗成粗实线。在计算机上进行修改时，可以用多义线命令Polyline直接画出不同宽度的直线、圆弧，但画圆环命令Donut在绘制不同宽度的圆轮廓线时，需要经常改变所绘制形体的宽度且进行多次重复计算。当需要改变零件图中某些形体的线条宽度时，由于多义线编辑命令Pedit每次只能改变一条线的宽度，所以修改的过程十分繁琐，而且有些形体是无法用多义线编辑命令Pedit来修改宽度的，如圆、椭圆及椭圆弧等。对圆的修改可以使用圆环命令Donut重新绘制；而对于椭圆及椭圆弧，则无相应编辑命令。所有这些，都给作图带来诸多不便，基于此，笔者利用VBA技术，结合自己对AutoCAD的应用体会，编写了能一次修改多个实体宽度的应用程序，大大方便了工程图形的绘制。仿照此法还可以对Pedit的其他命令选项进行扩展和集成。 

　　一、选择集的建立及管理 

　　为实现一次选择便可修改多个实体宽度的功能，采用VBA语句操作选择过程较为简单且容易理解。VBA中选择集的建立可以采用下述方法： 

　　Dim Selset as AcadSelectionSet　 */定义一个选择集"Selset" 

　　If Not IsNull(ThisDrawing.SelectionSets.Item("sset")) Then 

　　Set Selset = ThisDrawing.SelectionSets.Item("sset") 

　　Selset.Delete　 */如果选择集已存在，则删除 

　　End If 

　　Set Selset = ThisDrawing.SelectionSets.Add("sset")　*/添加选择集Selset 

　　　　 ThisDrawing是一个代表当前图形文件的程序术语，对独立的工程来说，ThisDrawing通常是指AutoCAD中的活动文档；而对内嵌的工程来说，ThisDrawing通常是指包含此工程的图形文档。 

　　选择集数据的建立模式通常有以下几种方法： 

　　1．Selset.Select Mode[， Point1][， Point2][， FilterType][， FilterData] */根据不同的Mode值来选择 

　　2．Selset.SelectAtPoint Point， FilterType， FilterData */选择通过给定点的对象 

　　3． Selset.SelectByPolygon Mode， PointsList， FilterType， FilterData */选择在给定网格内的对象 

　　4．Selset.SelectOnScreen　*/直接在屏幕上选择 

　　对于不同的图形来说，需要修改的实体不确定，故笔者采用的是最后一种方法，该方法可提供最大限度的灵活性。 

　　二、对直线宽度的修改 

　　由于多义线编辑命令Pedit提供了对直线宽度的编辑选项，因此对直线宽度的修改完全可以通过编程调用Pedit命令来实现，但由于本实例要实现预览、取消编辑的功能，笔者采用了在直线的位置上添加多义线并将该直线隐藏的方法，具体实现如下： 

　　Dim polyObj as AcadPolyline 

　　Set polyObj = ThisDrawing.ModelSpace.AddPolyline(pt)　*/生成多义线 

　　polyObj.SetWidth 0， ComBox.Text， ComBox.Text　*/改变起点、终点的宽度 

　　drawSet.Select acSelectionSetLast　*/将多义线添加到选择集"drawSet" 

　　polyObj.Layer = layerStr　*/使多义线与直线处于同一层，layerStr为直线所　　　　　　　　　　　在图层名称 

　　entObj.Visible = False　 */隐藏原直线 

　　三、对圆宽度的修改 

　　对于圆的修改，一种方式是用一条具有若干顶点的封闭多义线来近似模拟，绘制出的多义线失去了圆的基本特征，如圆心、半径等属性，一旦使用Explode命令，该多义线将被分解成若干条细实线；另一种方式是利用CAD内部的Donut命令，通过向AutoCAD的命令行发送Donut命令及其所需的内外直径的数值，修改后的圆环变为多义线，使用Explode命令时，该多义线将被分解成两个半圆弧，圆心、半径等属性不变。对比分析，我们使用了后一种方式，代码如下： 

　　If ent.ObjectName="AcDbCircle" then 

　　ThisDrawing.SendCommand　Chr(27) & "_donut" & vbCr & (2 * R -　 

　　Val(ComBox.Text)) & vbCr & (2 * R + Val(ComBox.Text)) & vbCr 

　　& (center(0) & "，" & center(1)) & NL　　 */发送画圆环命令 

　　Ent.Delete 

　　End if 

　　程序运行时遇到圆实体时，依次向AutoCAD的命令行输入区发送“DONUT”命令以及作圆环所需的内径、外径、圆心坐标等参数，结果是圆被取消，取而代之的是宽度为Val(ComBox.text)的圆环。 

　　四、对圆弧的修改，可以利用“三”中的后一种方式，只不过将“Donut”改为“Pline”，选择“Arc”模式。而对椭圆宽度的改变，则采用“三”中的前一种方式，即 

　　Dim plineObj as AcadPolyline 

　　Set plineObj=ThisDrawing.ModelSpace.AddPolyline(points) 

　　其中，Points为存放顶点坐标的一维数组，大小为顶点数的3倍。 

　　五、VBA应用程序扩展命令的载入 

　　若要使VBA应用程序只在一个图形文件中使用，可以将该程序内嵌到图形文件中，只要一打开这一图形文件，应用程序就会被自动装载。由于面向行业开发的应用程序一般都是通用的，若每次使用时都先进行加载，则比较麻烦，因此为了使开发的应用程序在创建新图形文件或启动AutoCAD时都可以使用，可以用Load Applications命令的Startup Suite选项，利用Add...按钮将该程序（后缀为.dvb的工程文件）加载进来。也可以用AutoLisp语言自定义函数S：：STARTUP，使其包含在acad.lsp或.mnl文件中，以后在启动AutoCAD或建立新文件时，应用程序将自动加载进来。以下是自定义的加载、运行、卸载函数并与定制的菜单项结合起来。 

　　(defun S::STARTUP() 

　　(defun C:LOADPEDIT() 

　　　　　　(command "_vbaload" "c:\polyline\pline.dvb") ) 

　　(defun C:RUNPEDIT() 

　　(command "_vbarun" "c:\polyline\pline.dvb!Module1.pline") ) 

　　(defun C:UNLOADPEDIT() 

　　(command "_vbaunload" "c:\polyline\pline.dvb") ) ) 

　　Module1.pline为pline.dvb工程对应的宏 

说明：使用其中的例子时将 */ 替换为 单引号
 VBA开发AutoCAD应用程序系列之一
VBA和VB应用程序之异同和相互移植
Microsoft VBA是一个面向对象的编程环境。它提供了与Visual Basic (VB)相似的丰富的开发能力。VBA和VB的最主要的区别在于VBA运行在与AutoCAD相同的处理空间，从而提供了一个具有AutoCAD智能的，非常快的编程环境。
VBA同时也提供与其它有VBA编程能力的应用程序的应用集成。这意味着AutoCAD利用其它应用程序的对象库可以控制其它应用程序，例如Microsoft Word和Excel。
VB是一个独立的开发版本，必须单独购买，但是它为AutoCAD VBA提供了许多附加的组件，象外部数据库引擎和报告输出能力等。
执行AutoCAD VBA有四点好处：
Visual Basic编程环境易学易用；
VBA运行在与AutoCAD相同的处理空间。使程序执行速度非常快。
对话框构建快速，高效。允许开发者快速地定制程序原形和收到设计的及时反馈。
工程可以单独存在，亦可嵌入在图形中。这就给了开发者在发行他们的应用程序时极大的灵活性。
要想用VB改写VBA的示例代码，你必须首先引用AutoCAD类型库。在VB里，从工程选单中选择引用选项激发引用对话框。从引用对话框，选择AutoCAD 版本15然后击OK。
下一步，在示例代码中用用户定义的指向当前活动文档的专门变量替代所有对ThisDrawing的引用。为了达到这个目的，定义一个AutoCAD应用程序的变量(acadApp)和一个当前文档的变量(acadDoc)。然后，设置应用程序变量(acadApp)为当前AutoCAD应用程序。
如果AutoCAD正在运行，GetObject方法就提取出AutoCAD应用程序对象。如果AutoCAD没有正在运行，一个错误就会发生，VB程序就应该捕捉并且清除它。CreateObject方法然后试图生成一个AutoCAD应用程序对象。如果成功，AutoCAD就启动；如果失败，一个显示错误描述的信息框就会出现。
下面的代码示例演示了VB如何连接到AutoCAD并且如何用Clear和Description属性处理错误信息变量Err。如果你的编程环境不支持这些属性，你需要适当地修改示例。
示例一：从Visual Basic 连接到AutoCAD
Sub ConnectToAcad()
Dim acadApp As AcadApplication
On Error Resume Next
Set acadApp = GetObject(, "AutoCAD.Application")
If Err Then
Err.Clear
Set acadApp = CreateObject("AutoCAD.Application")
If Err Then
MsgBox Err.Description
Exit Sub
End If
End If
MsgBox "Now running " + acadApp.Name + _
" version " + acadApp.Version
End Sub
下一步，设置文档变量为AutoCAD应用程序的文档对象。文档对象从应用程序对象的ActiveDocument属性返回。
示例二：获取AutoCAD文挡对象
Dim acadDoc as AcadDocument
Set acadDoc = acadApp.ActiveDocument
从这里开始，就可以使用acadDoc变量来引用当前的AutoCAD图形。
注意：当运行多个AutoCAD例程时，GetObject方法将返回在Windows运行对象表中的第一个AutoCAD例程。关于运行对象表(ROT)和GetObject方法请参看Microsoft Visual Basic相应文档。
示例三：VBA和VB代码对照
用VBA生成一条线
Sub AddLineVBA()
' This example adds a line
' in model space
Dim lineObj As AcadLine
Dim startPoint(0 To 2) As Double
Dim endPoint(0 To 2) As Double
' Define the start and end
' points for the line
startPoint(0) = 1
startPoint(1) = 1
startPoint(2) = 0
endPoint(0) = 5
endPoint(1) = 5
endPoint(2) = 0
' Create the line in model space
Set lineObj = ThisDrawing. _
ModelSpace.AddLine _
(startPoint, endPoint)
' Zoom in on the newly created line
ZoomAll
End Sub
用VB生成一条线
Sub AddLineVB()
On Error Resume Next
' Connect to the AutoCAD application
Dim acadApp As AcadApplication
Set acadApp = GetObject _
(, "AutoCAD.Application")
If Err Then
Err.Clear
Set acadApp = CreateObject _
("AutoCAD.Application")
If Err Then
MsgBox Err.Description
Exit Sub
End If
End If
' Connect to the AutoCAD drawing
Dim acadDoc As AcadDocument
Set acadDoc = acadApp.ActiveDocument
' Establish the endpoints of the line
Dim lineObj As AcadLine
Dim startPoint(0 To 2) As Double
Dim endPoint(0 To 2) As Double
startPoint(0) = 1
startPoint(1) = 1
startPoint(2) = 0
endPoint(0) = 5
endPoint(1) = 5
endPoint(2) = 0
' Create a Line object in model space
Set lineObj = acadDoc.ModelSpace.AddLine _
(startPoint, endPoint)
ZoomAll
End Sub
VBA开发AutoCAD应用程序系列之二 
VBA和其它API交互
VBA与Visual LISP交互：
通过VBARUN命令可以完成VLISP对VBA的调用。 
通过方法Sendcommand(command)可以完成VBA对VLISP的调用。
举例：
Sub VBASendCommand()
   ThisDrawing.SendCommand "_Circle" & vbCr & "2,2,0" & vbCr & "4" & vbCr
   ThisDrawing.SendCommand "_zoom" & vbCr & "a" & vbCr
   ThisDrawing.Regen acAllViewports
   MsgBox "A circle command has been sent to the command line of the current drawing."
End Sub
VBA与其它应用程序交互:
1. 建立对另一个应用程序对象库的引用
2. 建立另一个应用程序的例程
  Dim ExcelAppObj as Excel.Application
  Set ExcelAppObj = New Excel.Application
3. 结合对象编程
4. 退出
VBA使用DAO
1 建立Microsoft DAO对象库的参考. 
2 打开数据库. 
  Dim db As Database
  Set db = DBEngine.Workspaces(0).     
OpenDatabase("C:\X.MDB")
3  编制利用AutoCAD对象模型和DAO对象模型的代码.
VBA获取Windows APIs
声明Windows API：
1. 包含你想调用的进程的DLL名称
2. 在DLL中的进程名称
3. 在你的程序中你想要的名称
4. 进程所需参数
5. 返回值类型（如果你要调用的是一个函数）
调用Windows API：
参考 Win32api.txt
#If Win32 Then   
Declare Sub MessageBeep Lib "User32” Alias "SomeBeep" (ByVal N As Long)
#Else
    Declare Sub MessageBeep Lib "User” Alias "SomeBeep" (ByVal N As Integer)
#End If
Sub CallMyDll()
   Call MessageBeep(1) 
End Sub
